
This modification makes it possible to use the digitize facilities of the NMS8280 in 
black and white/grey. 
The modification is very simple and only requires the modification of one connection 
 
How to modify 
 
Remove the connection between pin 3 of IC303 (a V7030) and ground and connect 
this pin to pin 4 of connector AC (B/W). 
On the older version of the NMS8280 this requires the removalk of a wire, on the 
newer cutting the printconnection at pin 3. 
 
The pictures on the next pages show extracts of the relevant circuit diagram and the 
component side of the analog board. Read the NMS8280 service manual for more 
information. 



 
 



 
How to use 
 
When the computer is switched to black and white the ADC (Analog to Digital 
Converter) is switched to a 3x4 bit mode, 2x4 bits are connected.  
In black and white mode the values of the colors are equal: R=G=B. 
 
If the drawing program is set to grey then in screen 5 while digitizing 4 bits per pixel 
are used. In screen 7 there will be 2 pixels digitized (clocked) at the same time, total 
2x4 bits. Because this happens simultaneously this means in screen 7 every two 
adjacent pixels are equal. 
 



Program 
 
To use this modification the following program can be used: 
 
10 DIMR(15),G(15),B(15) 
11 OUT&HF6,&HFF 
20 FORN=1TO10:KEYN,"":NEXT 
24 ONERRORGOTO7000 
25 ONSTOPGOSUB26:STOPON:GOTO29 
26 POKE&HF3F6,&H10:RUN 
29 SETVIDEO0 
37 SCREEN5 
38 COLOR=NEW 
39 COLOR5,0,0 
40 CLS 
50 COLOR=(3,0,0,0) 
60 COLOR=(4,7,7,7) 
70 COLOR=(5,1,6,1) 
80 COLOR=(6,1,6,1) 
90 COLOR=(7,1,6,1) 
100 OPEN"grp:"AS#1 
110 LINE(60,10)-(188,20),8,B 
120 PSET(64,12) 
130 COLOR3 
140 PRINT#1,"SUPER DIGITISER" 
150 LINE(60,40)-(132,50),8,B 
160 PSET(64,42) 
170 COLOR4 
180 PRINT#1,"SCREEN 5" 
190 LINE(60,70)-(132,80),8,B 
200 PSET(64,72) 
210 COLOR5 
220 PRINT#1,"SCREEN 6" 
230 LINE(60,100)-(132,110),8,B 
240 PSET(64,102) 
250 COLOR6 
260 PRINT#1,"SCREEN 7" 
270 LINE(60,130)-(132,140),8,B 
280 PSET(64,132) 
290 COLOR7 
300 PRINT#1,"SCREEN 8" 
310 LINE(60,160)-(188,170),8,B 
320 PSET(64,162) 
330 COLOR3 
340 PRINT#1,"SUPER DIGITISER" 
350 GOSUB710 
360 SR=SC+1 
365 IFSR=8THEN2300:'>>>>>>>>>>>>>>>SCREEN8 
366 '********************************* 
370 COLOR=(2,0,0,6) 
380 COLOR=(3,0,0,0) 
390 COLOR=(4,7,7,7) 
400 COLOR=(5,1,6,1) 
410 COLOR=(6,1,6,1) 
420 COLOR=(7,1,6,1) 
430 CLS 
440 LINE(60,10)-(188,20),8,B 
450 PSET(64,12) 
460 COLOR2 
470 PRINT#1,"-PALLET CHOISE-" 



480 LINE(60,40)-(188,50),8,B 
490 PSET(64,42) 
500 COLOR4 
510 PRINT#1,"ORIGINAL PALLET" 
520 LINE(60,70)-(188,80),8,B 
530 PSET(64,72) 
540 COLOR5 
550 PRINT#1,"REAL B&W" 
560 LINE(60,100)-(188,110),8,B 
570 PSET(64,102) 
580 COLOR6 
590 PRINT#1,"REAL COLOR" 
600 LINE(60,130)-(188,140),8,B 
610 PSET(64,132) 
620 COLOR7 
630 PRINT#1,"REAL B&W&R" 
640 LINE(60,160)-(188,170),8,B 
650 PSET(64,162) 
660 COLOR3 
670 PRINT#1,"SUPER DIGITISER" 
680 GOSUB710 
690 PA=SC-4 
700 GOTO 2000 
701 '******************************* 
710 SC=4:XC=0 
720 ONSTRIGGOSUB990,990,990 
730 FORN=0TO2:STRIG(N)ON:NEXT 
740 N1=1:N2=7:ST=1 
750 FORCO=N1TON2STEPST 
760 COLOR=(3,0,0,CO) 
770 IFSTICK(0)<>0THEN 880 
780 A=PAD(12) 
790 B=PAD(14):IFB=1THENB=0 
800 XC=XC+ABS(B):IFXC<50THEN820 
810 GOTO 860 
820 FORN=1TO50:NEXT 
830 NEXT CO 
840 IFCO<5THEN740 
850 N1=7:N2=1:ST=-1:GOTO 750 
 
860 XC=0:IFB=ABS(B)THEN910 
870 GOTO 950 
880 IFSTICK(0)=5THEN910:'^ 
890 IFSTICK(0)=1THEN950:'V 
900 GOTO 820 
910 COLOR=(SC,1,6,1) 
920 SC=SC+1:IFSC=8THENSC=4 
930 COLOR=(SC,7,7,7) 
940 GOTO 820 
950 COLOR=(SC,1,6,1) 
960 SC=SC-1:IFSC=3THENSC=7 
970 COLOR=(SC,7,7,7) 
980 GOTO 820 
990 RETURN 995 
995 FORN=0TO2:STRIG(N)OFF:NEXT 
1000 RETURN 
1999 '*********************** 
2000 IF PA=0THEN2050:'ORIGINAL PALLET 
2010 IF PA=1THEN2500:'REAL B/W 
2020 IF PA=2THEN2100:'REAL COLOR 2100 
2030 IF PA=3THEN2200:'real b/w 2200 



2040 '******************************** 
2050 SCREENSR,1:COLOR=NEW:GOTO 3000 
2100 SCREEN SR,1 
2101 OUT&HF6,&B11101111 
2105 IFSR=6THEN2160 
2120 COLOR=NEW 
2150 GOTO 3000 
2160 COLOR=(0,0,0,0) 
2161 COLOR=(1,7,0,0) 
2162 COLOR=(2,0,7,0) 
2163 COLOR=(3,0,0,7) 
2165 GOTO 3000 
2200 SCREENSR,1:OUT&HF6,&B11101111 
2209 FORN=0TO15 
2210 COLOR=(N,N/2,N/2,N/2) 
2220 NEXT 
2221 FORN=1TO14STEP2 
2222 COLOR=(N,N/2+1,N/2,N/2) 
2223 NEXT 
2230 GOTO 3000 
2300 SCREEN SR:GOTO 3000 
2500 'real b/w 
2510 SCREENSR,1 
2520 OUT&HF6,&B11101111 
2525 IFSR=6THEN2580 
2530 FORN=0TO15 
2540 COLOR=(N,N/2,N/2,N/2) 
2560 NEXT 
2570 GOTO 3000 
2580 COLOR=(0,0,0,0) 
2590 COLOR=(1,2,2,2) 
2600 COLOR=(2,4,4,4) 
2610 COLOR=(3,7,7,7) 
2620 GOTO 3000 
3000 ONKEYGOSUB3150,3210,3270:'** save ** load ** edit 
3010 FORN=1TO3:KEY(N)OFF:NEXT 
3020 SETVIDEO3 
3030 CLS 
3031 A=1 
3035 FORN=1TO100:NEXT 
3040 IFNOT(STRIG(0)ORSTRIG(1))THEN3040 
3050 IFSR<>8THENCOLOR,,15ELSECOLOR,,255 
3060 COPY SCREEN 
3070 FORN=1TO5:NEXT 
3080 SETVIDEO 1 
3090 COLOR,,0 
3095 IFA=0THEN3100 
3096 FORN=1TO100:NEXT:A=0 
3100 IFSTRIG(0)ORSTRIG(1)THEN3050 
3110 FORN=1TO2:KEY(N)ON:NEXT 
3115 IFSR<>8THENKEY(3)ON 
3120 IFSTRIG(0)ORSTRIG(1)THEN3010 
3130 GOTO 3120 
3140 '************************** 
3150 FORN=1TO3:KEY(N)OFF:NEXT 
3155 MO$="SAVE" 
3160 GOSUB 5910:'>>>>>>> naam 
3175 IFNA$=""THEN3190 
3180 IFSR=8THENBSAVE NA$+".pic",0,&HD3FF,S:GOTO 3190 
3181 IFSR=7THENBSAVE NA$+".tm7",0,&HD3FF,S:BSAVE 
NA$+".pm7",&HFA80,&HFA9F,S:GOTO 3190 



3182 IFSR=6THENBSAVE NA$+".tm6",0,&H69FF,S:BSAVE 
NA$+".pm6",&H7680,&H769F,S:GOTO 3190 
3183 BSAVE NA$+".tm5",0,&H69FF,S:BSAVE NA$+".pm5",&H7680,&H769F,S 
3190 FORN=1TO2:KEY(N)ON:NEXT 
3195 IFSR<>8THENKEY(3)ON 
3200 RETURN 
3210 FORN=1TO3:KEY(N)OFF:NEXT 
3211 MO$="LOAD" 
3220 GOSUB 5910:'>>>>>>> naam 
3225 IFNA$=""THEN3250 
3230 IFSR=8THENBLOAD NA$+".pic",S:GOTO 3250 
3231 IFSR=7THENBLOAD NA$+".tm7",S:BLOAD NA$+".pm7",S:GOTO 3240 
3232 IFSR=6THENBLOAD NA$+".tm6",S:BLOAD NA$+".pm6",S:GOTO 3240 
3233 BLOAD NA$+".tm5",S:BLOAD NA$+".pm5",S 
3240 COLOR=RESTORE 
3250 FORN=1TO2:KEY(N)ON:NEXT 
3251 IFSR<>8THENKEY(3)ON 
3260 RETURN 
3270 FORN=1TO3:KEY(N)OFF:NEXT 
3280 GOSUB5000 
3290 ONKEYGOSUB3150,3210,3270 
3300 FORN=1TO3:KEY(N)ON:NEXT 
3310 RETURN 
5000 NR=15:KA=15:XC=0 
5010 IFSR=6THENNR=3:KA=3 
5020 SP$=STRING$(6,CHR$(&HFF)) 
5030 FORN=0TONR 
5040 SPRITE$(N)=SP$ 
5050 PUTSPRITEN,(10,N*13),N,N 
5060 COLORSPRITE$(N)=CHR$(KA)+STRING$(4,CHR$(N))+CHR$(KA) 
5070 NEXT 
5080 GOSUB5750 
5090 ONKEYGOSUB5570,5630,5700 
5100 FORN=1TO3:KEY(N)ON:NEXT 
5110 PUTSPRITESP,(0,220),,SP 
5120 FORN=1TO50:NEXT 
5130 PUTSPRITESP,(10,SP*13),,SP 
5140 A=STICK(0):GOSUB 5220 
5150 A=PAD(12) 
5160 B=PAD(14):IFB=1THENB=0 
5170 XC=XC+ABS(B):IFXC<25THEN5190 
5180 GOSUB5250 
5190 A$=INKEY$:IFA$<>""THENGOSUB5310 
5200 FORN=1TO50:NEXT 
5210 GOTO 5110 
5220 IF A=5THEN5270 
5230 IF A=1THEN5290 
5240 RETURN 
5250 IFB=ABS(B)THENXC=0:GOTO 5270 
5260 XC=0:GOTO 5290 
5270 SP=SP+1:IFSP>NRTHENSP=0 
5280 RETURN 
5290 SP=SP-1:IFSP<0THENSP=NR 
5300 RETURN 
5310 IFA$="r"THENGOSUB5380 
5320 IFA$="R"THENGOSUB5410 
5330 IFA$="g"THENGOSUB5440 
5340 IFA$="G"THENGOSUB5470 
5350 IFA$="b"THENGOSUB5500 
5360 IFA$="B"THENGOSUB5530 
5370 RETURN 



5380 R(SP)=R(SP)+1:IFR(SP)>7THENR(SP)=7 
5390 COLOR=(SP,R(SP),G(SP),B(SP)) 
5400 RETURN 
5410 R(SP)=R(SP)-1:IFR(SP)<0THENR(SP)=0 
5420 COLOR=(SP,R(SP),G(SP),B(SP)) 
5430 RETURN 
5440 G(SP)=G(SP)+1:IFG(SP)>7THENG(SP)=7 
5450 COLOR=(SP,R(SP),G(SP),B(SP)) 
5460 RETURN 
5470 G(SP)=G(SP)-1:IFG(SP)<0THENG(SP)=0 
5480 COLOR=(SP,R(SP),G(SP),B(SP)) 
5490 RETURN 
5500 B(SP)=B(SP)+1:IFB(SP)>7THENB(SP)=7 
5510 COLOR=(SP,R(SP),G(SP),B(SP)) 
5520 RETURN 
5530 B(SP)=B(SP)-1:IFB(SP)<0THENB(SP)=0 
5540 COLOR=(SP,R(SP),G(SP),B(SP)) 
5550 RETURN 
5560 '*********************************** 
5570 '************** save pallet 
5580 FORN=1TO3:KEY(N)OFF:NEXT 
5585 MO$="SAVE" 
5590 GOSUB 5910:'>>>>>>>>>naam 
5600 IFNA$=""THEN5620 
5610 IFSR=7THENBSAVE NA$+".pm7",&HFA80,&HFA9F,S:GOTO 5620 
5611 IFSR=6THENBSAVE NA$+".pm6",&H7680,&H769F,S:GOTO 5620 
5612 BSAVE NA$+".pm5",&H7680,&H769F,S 
5620 RETURN 5080 
5630 '************** load pallet 
5640 FORN=1TO3:KEY(N)OFF:NEXT 
5645 MO$="LOAD" 
5650 GOSUB 5910:'>>>>>>>>>naam 
5660 IFNA$=""THEN5690 
5670 IFSR=7THENBLOAD NA$+".pm7",S:GOTO 5680 
5671 IFSR=6THENBLOAD NA$+".pm6",S:GOTO 5680 
5672 BLOAD NA$+".pm5",S 
5680 COLOR=RESTORE 
5690 RETURN 5030 
5700 '************** exit 
5710 FORN=1TO3:KEY(N)OFF:NEXT 
5714 DEFUSR=&H69:A=USR(0) 
5720 RETURN5730 
5730 RETURN 
5740 '********************************** 
5750 ' 
5760 IFSR=5THENVA=&H7680 
5770 IFSR=6THENVA=&H7680 
5780 IFSR=7THENVA=&HFA80 
5790 M=VA 
5800 FORN=0TO15 
5810 A$="0000000"+BIN$(VPEEK(M)) 
5820 A$=RIGHT$(A$,8) 
5830 B(N)=VAL("&b"+RIGHT$(A$,4)) 
5840 R(N)=VAL("&b"+LEFT$(A$,4)) 
5850 M=M+1 
5860 G(N)=VPEEK(M) 
5870 M=M+1 
5880 NEXT 
5890 RETURN 
5900 '************************************** 
5910 N=15:IFSR=8THENN=255 



5915 COLORN,1 
5920 COPY(80,100)-(220,140),0TO(80,100),1 
5930 LINE(80,130)-(220,140),1,BF 
5931 LINE(80,100)-(120,110),1,BF 
5940 LINE(80,130)-(220,140),N,B 
5941 LINE(80,100)-(120,110),N,B 
5945 PSET(82,102) 
5949 PRINT#1,MO$ 
5950 PSET(82,132) 
5960 PRINT#1,"FILENAME:"; 
5970 NA$="" 
5980 FORX=154TO154+8*8 STEP8 
5990 PRESET(X,132) 
6000 A$=INPUT$(1) 
6010 IFA$=CHR$(&H1B)THENCOPY(80,100)-
(220,140),1TO(80,100),0:NA$="":RETURN 
6020 IFA$=CHR$(&HD)THEN6140 
6030 IFA$=CHR$(&H8)THEN 6100 
6040 IFA$=CHR$(&H1D)THEN 6100 
6050 IF ASC(A$)<=&H20THEN5990 
6060 IF ASC(A$)>=&H7BTHEN5990 
6070 IFX>154+7*8THEN6140 
6080 NA$=NA$+A$:PRINT#1,A$:NEXTX 
6090 GOTO 6140 
6100 A$=" " 
6110 X=X-8:IFX<=154THENX=154:NA$=" " 
6120 PRESET(X,132):PRINT#1,A$ 
6130 NA$=LEFT$(NA$,LEN(NA$)-1):GOTO 5990 
6140 COPY(80,100)-(220,140),1TO(80,100),0 
6145 FORN=1TO20:NEXT 
6150 RETURN 
7000 '************************************************ 
7001 N=15:IFSR=8THENN=255 
7002 COLORN,1 
7003 STOP OFF 
7010 COPY(80,130)-(220,140),0TO(80,130),1 
7020 LINE(80,130)-(220,140),1,BF 
7030 LINE(80,130)-(220,140),N,B 
7035 ER$="" 
7040 IF ERR=66THENER$="   DISK FULL" 
7050 IF ERR=69THENER$="DISK I/O ERROR" 
7060 IF ERR=70THENER$=" DISK OFFLINE" 
7070 IF ERR=68THENER$="WRITE PROTECTED" 
7080 IF ERR=53THENER$="FILE NOT FOUND" 
7090 IF ERR=67THENER$=" TO MANY FILES" 
7100 IF ERR=61THENER$=" BAD FILE MODE" 
7110 IF ERR=56THENER$=" BAD FILE NAME" 
7120 IF ERR=60THENER$="    BAT FAT" 
7130 IFER$=""THENER$=" PROGRAM ERROR" 
7140 PSET (82,132):PRINT#1,ER$ 
7150 IFINKEY$<>""THEN7150 
7160 IFINKEY$=""THEN7160 
7170 COPY(80,130)-(220,140),1TO(80,130),0 
7171 FORN=1TO100:NEXT 
7175 DEFUSR=&H156:A=USR(0) 
7180 STOP ON 
7190 IFERL>3160ANDERL<3250THENRESUME3250 
7200 RESUME NEXT 
  


